Subroutine/Dictionaries for Date an Account Went Negative

All items are loaded at the Real TCL prompt, not XTCL if using MESA.  If you are not able to get to Real TCL, please contact me and I will give you some assistance.

The subroutine listed below is stored in a directory called SUBR.BP for no other reason than this is what I decided to call it.  If I have performed work for your CU in the past and have loaded a subroutine, it would have been loaded in this file.  To determine if this file exists, type LIST SUBR.BP and if items list, you will not have to create the file as in step 1.

Follow each step below by using a copy/paste from this document to your Telnet session.

1. At TCL, type CREATE-FILE DIR SUBR.BP
2. After the file is created, type AE SUBR.BP NEG.DATE
3. Assuming this name has not been used, the screen will display:


Top of New “NEG.DATE” in “SUBR.BP”

4. Type I for insert to get to line 001= and then copy/paste the following lines:

$BASICTYPE "U"

SUBROUTINE NEG.DATE(ANSWER,ID,BAL)

* to find when a share act when negative

* bp 11/02/10

***********************************************************

ANSWER="";TDATE=""

IF BAL GE "0" THEN GO DONE

OPEN '','TRAN' TO TRAN ELSE GO DONE

OPEN '','TRAN.EXT' TO TRANEXT ELSE GO DONE

READ RECORD FROM TRAN,ID ELSE GO DONE

*first loop through TRAN in reverse date order

*to add transaction amts back in until the balance goes positive

*or until you run out of TRAN attributes in the record

HOWMANY=DCOUNT(RECORD,@VM)

FOR JJ=HOWMANY TO 17 STEP -1

  TCODE=RECORD<JJ,1,1>

  IF TCODE < 15 THEN GO NEXTJJ

  AMT=RECORD<JJ,3,1>

  TDATE=RECORD<JJ,2,1>

  BEGIN CASE

    CASE TCODE < "50" OR TCODE = "87"

      AMT=AMT * -1

    CASE 1

      NULL

  END CASE

  BAL=BAL+AMT

  IF BAL GE "0" THEN GO DONE

NEXTJJ:

NEXT JJ

*now check the TRAN.EXT records (if any)

NUMEXTS=DCOUNT(RECORD<3,1>,@SM)

IF NUMEXTS < 1 THEN GO DONE

FOR LL=NUMEXTS TO 1 STEP -1

  EXTID=ID:"/":LL

  READ RECORD FROM TRANEXT,EXTID ELSE GO DONE

  HOWMANYLINES=DCOUNT(RECORD,@AM)

  FOR KK= HOWMANYLINES TO 1 STEP -1

    TCODE=RECORD<KK,1,1>

    IF TCODE < 15 THEN GO NEXTKK

    AMT=RECORD<KK,3,1>

    TDATE=RECORD<KK,2,1>

    BEGIN CASE

      CASE TCODE < "50" OR TCODE = "87"

        AMT=AMT * -1

      CASE 1

        NULL

    END CASE

    BAL=BAL+AMT

    IF BAL GE "0" THEN GO DONE

NEXTKK:

  NEXT KK

NEXT LL

***********************************************************

DONE:

ANSWER=TDATE

RETURN

5. When the copy is complete, hit RETURN to get to the prompt line *--: and type FI to file item
6. At TCL, type BASIC SUBR.BP NEG.DATE to compile the subroutine
7. At TCL, type CATALOG SUBR.BP NEG.DATE LOCAL to catalog the subroutine
8. Create the following dictionaries at Real TCL:
AE DICT TRAN NEG.DATE
I

SUBR("NEG.DATE",@ID,BAL)

D2

DATE NEG

9R
S
AE DICT TRAN NEG.DAYS

I

@DATE – NEG.DATE + 1 ; IF BAL < "0" THEN @1 ELSE ""
MD0,

DAYS NEG

8R

S

(Note: The following dictionary may already be created in the TRAN file on your system and is needed by the subroutine)

AE DICT TRAN BAL

D

1

MD2,

BALANCE

12R

S
Sample Recall Sentences:

SELECT S10.INDEX

SELECT TRAN WITH BAL LE "-0.01"
LIST TRAN NEG.DATE NEG.DAYS

